



GrowTechie

DAX: From Zero To Hero

Swipe Right



@ramakrushnamohapatra



The DAX Language

1. Language Of BI

- Power BI
- Analytics Service Tabular
- Power Pivot

1. DAX Is Simple, But Its not easy

1. New Programming Concepts and Patterns



What Is DAX ?

1. Programming language
 - Power BI
 - Analysis Services Tabular
 - Power Pivot
1. Resembles Excel
 - Because it was born with PowerPivot
 - Important differences
 - No concept of «row» and «column»
 - Different type system
1. Many new functions
2. Designed for data models and business



Functional Language

DAX is a functional language, the execution flows with function calls, here is an example of a DAX formula.

```
=SUMX (FILTER (  
VALUES ( 'Date'[Year] ),  
'Date'[Year] < 2005  
),  
IF (  
'Date'[Year] >= 2000,  
[Sales Amount] * 100,  
[Sales Amount] * 90  
))
```



If it is not formatted, it is not DAX

Code formatting is of paramount importance in DAX.

```
=SUMX(FILTER(VALUES('Date'[Year]),'Date'[Year]<2005),IF('Date'[Year]>=2000,[Sales  
Amount]*100,[Sales Amount]*90)
```

```
=SUMX ( FILTER (  
VALUES ( 'Date'[Year] ),  
'Date'[Year] < 2005  
) ,  
IF ( 'Date'[Year] >= 2000,  
[Sales Amount] * 100,  
[Sales Amount] * 90  
))
```



DAX Data Types

1. Numeric types

- Integer (64 bit)
- Decimal (floating point)
- Currency (money)
- Date (DateTime)
- TRUE / FALSE (Boolean)

2. Other types

- String
- Binary Objects



DAX Type Handling

1. Operator Overloading

- Operators are not strongly typed
- The result depends on the inputs

1. Example:

- "4" + "8" = 12
- 4 & 8 = "48"

1. Conversion happens when needed

- Pay attention to undesired conversions



DateTime

1. Floating point value
1. Integer part
 - Number of days after December, 30, 1899
1. Decimal part
 - Seconds: $1 / (24 * 60 * 60)$
1. DateTime Expressions
 - Date + 1 = The day after
 - Date - 1 = The day before



Calculated Columns

1. Columns computed using DAX
1. Always computed row by row
1. Product[Price] means
 - The value of the Price column (explicit)
 - In the Product table (explicit, optional)
 - For the current row (implicit)
 - Different for each row



Column References

1. The general format to reference a column
 - 'TableName'[ColumnName]
1. Quotes can be omitted
 - If TableName does not contain spaces
 - Do it: omit quotes if there are no spaces in table name
1. TableName can be omitted
 - Current table is searched for ColumnName
 - Don't do it, harder to understand



Measures

1. Written using DAX
1. Do not work row by row
1. Instead, use tables and aggregators
1. Do not have the «current row» concept
1. Examples
 - GrossMargin
 - is a calculated column
 - but can be a measure, too
 - GrossMargin %
 - must be a measure



Naming Convention

1. Measures should not belong to a table
 - Avoid table name
 - [Margin%] instead of Sales[Margin%]
 - Easier to move to another table
 - Easier to identify as a measure

1. Use this syntax when to reference:
 - Columns → Table[Column]
 - Measures → [Measure]



Measures vs Calculated Columns

1. Use a column when
 - Need to slice or filter on the value

1. Use a measure
 - Calculate percentages
 - Calculate ratios
 - Need complex aggregations

1. Space and CPU usage
 - Columns consume memory in the model
 - Measures consume CPU at query time



Aggregation Functions

1. Useful to aggregate values
 - SUM
 - AVERAGE
 - MIN
 - MAX

1. Aggregate only one column
 - $\text{SUM}(\text{Orders[Price]})$
 - $\text{SUM}(\text{Orders[Price]} * \text{Orders[Quantity]})$



The «X» Aggregation Functions

1. Iterators: useful to aggregate formulas
 - SUMX
 - AVERAGEX
 - MINX
 - MAXX
1. Iterate over the table and evaluate the expression for each row
1. Always receive two parameters
 - Table to iterate
 - Formula to evaluate for each row



Example Of SUMX

For each row in the Sales table, evaluates the formula, then sum up all the results.
Inside the formula, there is a «current row».



SUM or SUMX?

Actually, SUM is nothing but syntax sugar for SUMX

```
--  
-- This is the compact format for a SUM  
--  
SUM ( Sales[Quantity] )  
--  
-- Internally, this is translated into  
--  
SUMX (  
Sales,  
Sales[Quantity]  
)
```



IN Operator

1. Verify if the result of an expression is included in a list of values:

```
Customer[State] IN { "WA", "NY", "CA" }
```

1. It would require multiple **OR** conditions otherwise:

```
Customer[State] = "WA"  
|| Customer[State] = "NY"  
|| Customer[State] = "CA"
```



The DIVIDE Function

Divide is useful to avoid using IF inside an expression to check for zero denominators. It is also faster than using IF.

```
IF (  
    Sales[SalesAmount] <> 0,  
    Sales[GrossMargin] / Sales[SalesAmount],  
    0)
```

You can write it better with DIVIDE

```
DIVIDE (  
    Sales[GrossMargin],  
    Sales[SalesAmount],  
    0)
```



Using Variables

Very useful to avoid repeating subexpressions in your code.

VAR

```
TotalQuantity = SUM ( Sales[Quantity] )
```

RETURN

```
IF (  
    TotalQuantity > 1000,  
    TotalQuantity * 0.95,  
    TotalQuantity * 1.25  
)
```



Relational Functions

1. RELATED

- Follows relationships and returns the value of a column

1. RELATEDTABLE

- Follows relationships and returns all the rows in relationship with the current one

1. It doesn't matter how long the chain of relationships is

- All the relationships must be in the same direction



Some functions return tables instead of values

Table Functions



Table Functions

1. Basic functions that work and/or return full tables

- FILTER
- ALL
- VALUES / DISTINCT
- RELATEDTABLE
- ADDCOLUMNS / SUMMARIZE

1. They are used very often in DAX – their knowledge is required
2. Their result is often used in other functions
3. They can be combined together to form complex expressions
4. We will discover many other table functions later in the course



Filtering A Table

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	120	1

```
SUMX (  
    FILTER ( Orders,  
        Orders[Price] > 1  
    ),  
    Orders[Quantity] * Orders[Price]  
)
```



Channel

- Internet
- Store

	Color	Large	Small	Total
Green	192			192
Red	112	160		272



The FILTER Function

1. FILTER

- Adds a new condition
- Restricts the number of rows of a table
- Returns a table
- Can be iterated by an «X» function

1. Needs a table as input
2. The input can be another FILTER



Ignoring Filters

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

```
SUMX (  
    ALL ( Orders ),  
    Orders[Quantity] * Orders[Price]  
)
```

ANALYSIS

Channel

- Internet
- Store

	Color	Large	Small	Total
Green	749	749	749	749
Red	749	749	749	749



The ALL Function

1. ALL

- Returns all the rows of a table
- Ignores any filter
- Returns a table
- That can be iterated by an «X» function

1. Needs a table as input
2. Can be used with a single column

- **ALL (Customers[CustomerName])**
- The result contains a table with one column



ALL With Many Columns

Returns a table with all the values of all the columns passed as parameters.

```
COUNTROWS (  
    ALL (  
        Orders[Channel],  
        Orders[Color],  
        same table  
        Orders[Size]  
    )  
)
```

Columns of the



Mixing Filters


1. Table functions can be mixed
1. Each one requires a table
1. Each one returns a table
1. FILTER (ALL (Table), Condition)
 - Puts a filter over the entire table
 - Ignores the current filter context



Mixing Filters

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

```
SUMX (  
    FILTER (   
        ALL( Orders ),  
        Orders[Channel] = "Internet"  
    ),  
    Orders[Quantity] * Orders[Price]  
)
```



	Color	Large	Small	Total
Green	592	592	592	592
Red	592	592	592	592



DISTINCT

Returns the unique values of a column, only the ones visible in the current filter context.

```
NumOfProducts :=  
    COUNTROWS (  
        DISTINCT ( Product[ProductCode] )  
    )
```



How Many Values For A Column?

Returns the unique values of a column, only the ones visible in the current filter context.

The diagram illustrates a relationship between two tables. On the left, a table shows rows of Amount and ProductId with green checkmarks in the last column. A blue arrow labeled "Relationship" points from the ProductId column of this table to the ProductId column of the table on the right. The right table lists products by ProductId: 1 (Coffee), 2 (Pasta), 3 (Tomato), and two blank rows. A red arrow points from the right table towards the bottom right corner.

Amount	ProductId	
25.00	1	<input checked="" type="checkbox"/>
12.50	2	<input checked="" type="checkbox"/>
2.25	3	<input checked="" type="checkbox"/>
2.50	3	<input checked="" type="checkbox"/>
14.00	4	
16.00	5	

ProductId	Product
1	Coffee
2	Pasta
3	Tomato
BLANK	BLANK
BLANK	BLANK

Tables targets of a relationship might contain an additional blank row, created by DAX to guarantee referential integrity.



VALUES

Returns the unique values of a column, only the ones visible in the current filter context, including the additional blank row if it is visible in the filter context.

```
NumOfProducts :=  
    COUNTROWS (  
        VALUES ( Product[ProductCode] )  
    )
```



Counting Different Values

ALL returns the additional blank row, if it exists. ALLNOBLANKROW omits it.

Product	CountRowsAll	CountRowsDistinct	CountRowsValues	CountRowsAllNoBlankRow
	4		1	3
Coffee	4	1	1	3
Pasta	4	1	1	3
Tomato	4	1	1	3
Total	4	3	4	3

Note the difference among

- o DISTINCT
- o VALUES
- o ALL
- o ALLNOBLANKROW



ALLSELECTED

ALLSELECTED returns the elements of a table as they are visible outside of the current visual, be either a pivot table in Excel or a visual in Power BI.

```
Pct All =  
DIVIDE (  
    [Sales Amount],  
    SUMX (  
        ALL ( Sales ),  
        Sales[Quantity] * Sales[Net Price]  
    )  
)
```

```
Pct AllSel =  
DIVIDE (  
    [Sales Amount],  
    SUMX (  
        ALLSELECTED ( Sales ),  
        Sales[Quantity] * Sales[Net Price]  
    )  
)
```

Category	Category	Sales Amount	Pct All	Pct AllSel
■ Audio	Cameras and camcorders	842,849,210.26	22.15%	47.25%
■ Cameras and camcorders	Computers	842,569,053.80	22.14%	47.24%
■ Cell phones	Audio	51,553,689.31	1.35%	2.89%
■ Computers	Games and Toys	46,769,456.76	1.23%	2.62%
■ Games and Toys	Total	1,783,741,410.13	46.87%	100.00%
■ Home Appliances				
■ Music, Movies and Audio...				
■ TV and Video				



RELATEDTABLE

Returns a table with all the rows related with the current one.

```
NumOfProducts =
```

```
COUNTROWS (
```

```
RELATEDTABLE ( Product )
```

```
)
```



ADDCOLUMNS

Adds one or more columns to a table expression, keeping all existing columns.
It is an iterator, therefore you can access columns of the iterated table

ColorsAndSales

ADDCOLUMNS (

VALUES ('Product'[Color]), "Sales",

SUMX (

RELATEDTABLE (Sales),

Sales[Quantity] * Sales[Net Price]

))



Tables with one row and one column

When a table contains ONE row and ONE column, you can treat it as a scalar value.

Sel Category :=

"You selected: " &

IF (
HASONEVALUE ('Product Category'[Category]),

VALUES ('Product Category'[Category]),

"Multiple values"
)



SELECTEDVALUE

SELECTEDVALUE is a convenient function that simplifies retrieving the value of a column, when only one value is visible.

```
SELECTEDVALUE (
    'Product Category'[Category],
    "Multiple values"
)
```

Equivalent to:

```
IF (
    HASONEVALUE ( 'Product Category'[Category] ),
    VALUES ( 'Product Category'[Category] ),
    "Multiple values"
)
```

Table variables

A variable can contain either a scalar value or a table. Using table variables greatly helps in splitting complex expressions.

```
VAR  
SalesGreaterThan10 = FILTER ( Sales, Sales[Quantity] > 10 )  
  
RETURN  
  
SUMX (  
FILTER (   
SalesGreaterThan10,  
RELATED ( Product[Color] ) = "Red"  
,  
  
Sales[Amount]  
)
```



Let us take a look at how DAX works

Evaluation Contexts



Evaluation contexts

- 1. Evaluation contexts are the pillars of DAX
- 1. Simple concepts, hard to learn
- 1. At the beginning, they looks very easy
- 1. Using them, complexity arises
- 1. The devil is in the details



What is an evaluation context?

TotalSales := SUMX (Sales, Sales[Quantity] * Sales[Net Price])

The screenshot shows a Power BI interface. On the left, a calculated column named 'TotalSales' is displayed with the value '\$29,358,677.22'. A large blue arrow points from this value to a pivot table on the right. The pivot table has 'Row Labels' set to 'TotalSales' and shows sales data categorized by color. The data is as follows:

Color	TotalSales
Black	\$8,838,411.96
Blue	\$2,279,096.28
Multi	\$106,470.74
NA	\$435,116.69
Red	\$7,724,330.52
Silver	\$5,113,389.08
White	\$5,106.32
Yellow	\$4,856,755.63
Grand Total	\$29,358,677.22

A callout box with a blue border and white text explains: "Numbers are sliced by color, i.e. the formula is NOT computing sum of sales, it is computing it for only a subset of the data model".

The value of a formula depends on its context



Filter context in a pivot table

The diagram illustrates the context of filters applied to a PivotTable. A blue box labeled "PivotTable Filter" points to the "Gender" dropdown menu, which is set to "FEMALE". Another blue box labeled "Columns" points to the column headers "Bachelor's", "Graduate Degree", "High School", "Partial College", "Partial High School", and "Grand Total". A blue box labeled "Rows" points to the row labels "Black", "Blue", "Multi", and "NA". A blue box labeled "Slicers" points to the "ProductModel" slicer on the left, which lists items like "All-Purpose Bike St...", "Bike Wash", "Classic Vest", "Cycling Cap", "Fender Set - Mount...", "Half-Finger Gloves", "Hitch Rack - 4 Bike", and "HL Mountain Tire". Arrows show how the filter "FEMALE" applies to the "Graduate Degree" column, and how the "ProductModel" slicer applies to the "Half-Finger Gloves" row.

Row Labels	Bachelor's	Graduate Degree	High School	Partial College	Partial High School	Grand Total
Black	\$5,142.90	\$3,012.27	\$3,012.27	\$4,481.67	\$1,297.97	\$16,947.08
Blue	\$5,143.50	\$3,238.50	\$2,984.50	\$4,826.00	\$1,270.00	\$17,462.50
Multi	\$2,876.80	\$1,717.09	\$1,672.14	\$2,813.87	\$737.18	\$9,817.08
NA	\$8,090.95	\$5,444.01	\$3,913.36	\$6,996.16	\$1,984.74	\$26,429.22
Grand Total	\$21,254.15	\$13,411.87	\$11,582.27	\$19,117.70	\$5,289.89	\$70,655.88



Example of a filter context

The diagram illustrates the flow of filter context in Power BI. A blue box highlights a subset of data in a table, which is then connected by a red line to a blue box containing two dropdown filters: "Channel" (set to Internet) and "City" (set to New York). This context is then applied to a pivot table below. The pivot table has "Row Labels" set to "Large" and "Small". The "Large" row shows values 64 (Green) and 128 (Red), with a yellow arrow pointing to their sum of 192. The "Small" row shows values 16 (Green) and 32 (Red), with a yellow arrow pointing to their sum of 48. The "Grand Total" cell contains 240, with a blue arrow pointing to it from the bottom right.


City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

	A	B	C	D	E	F	G
1							
2							
3							
4							
5							
6							
7							
8							
9							
10							
11							
12							
13							

	Sum of Quantity	Column Labels		
Row Labels		Large	Small	Grand Total
Green		64	128	192
Red		16	32	48
Grand Total	80	160		240



Filter context in Power BI





Filter context

1. Defined by
 - Row Selection
 - Column Selection
 - Report Filters
 - Slicers Selection

1. Rows outside of the filter context
 - Are not considered for the computation

1. Defined automatically by the client, tool

1. Can also be created with specific functions



Row context

1. Defined by
 - Calculated column definition
 - Defined automatically for each row
 - Row Iteration functions
 - SUMX, AVERAGEX ...
 - All «X» functions and iterators
 - Defined by the user formulas
2. Needed to evaluate column values, it is the concept of “current row”



SUMX (Orders, Orders[Quantity]*Orders[Price])

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

$$16 \times 7 = 112$$

$$32 \times 5 = 160$$

$$64 \times 3 = 192$$

$$128 \times 1 = 128$$

SUM = 592

Channel	Color	Large	Small	Total
Internet	Green	192	128	320
Internet	Red	112	160	272
	Total	304	288	592

Channel
Internet



There are always two contexts

1. Filter context

- Filters tables
- Might be empty
- All the tables are visible
- But this never happens in the real world

1. Row context

- Iterates rows
- For the rows active in the filter context
- Might be empty
- There is no iteration running

1. Both are «evaluation contexts»



Filtering a table

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

```
SUMX (  
    FILTER ( Orders,  
        Orders[Price] > 1  
    ),  
    Orders[Quantity] * Orders[Price]  
)
```

Channel	Color	Large		Small	Total
		192	160	272	464
■ Internet	Green	192	160	272	464
□ Store	Red	112	160	272	464
	Total	304	160	464	





Ignoring filters

City	Channel	Color	Size	Quantity	Price
Paris	Store	Red	Large	1	15
Paris	Store	Red	Small	2	13
Torino	Store	Green	Large	4	11
New York	Store	Green	Small	8	9
	Internet	Red	Large	16	7
	Internet	Red	Small	32	5
	Internet	Green	Large	64	3
	Internet	Green	Small	128	1

```
SUMX (  
    ALL ( Orders ),  
    Orders[Quantity] * Orders[Price]  
)
```

Channel		Color	Large		Small	Total
			Internet	Store	749	
Internet	Green		749	749	749	749
Internet	Red		710	710	710	710





Using RELATED in a row context

Starting from a row context, you can use RELATED to access columns in related tables.

```
SUMX (  
Sales,  
Sales[Quantity]  
* RELATED ( Products[ListPrice] )  
* RELATED ( Categories[Discount] )  
)
```

You need RELATED because the row context is iterating the Sales table



Nesting row contexts

Row contexts can be nested, on the same or on different tables.

```
SUMX (  
    Categories,  
    SUMX (  
        RELATEDTABLE ( Products ),  
        SUMX (  
            RELATEDTABLE ( Sales )  
  
                ( Sales[Quantity] * Products[ListPrice] )  
                * Categories[Discount]  
        )  
    )  
)
```

- Three row contexts:
- Categories
 - Products of category
 - Sales of product



Ranking by price

- 1. Create a calculated column
- 1. Ranks products by list price
- 1. Most expensive product is ranked 1

ProductKey	ProductName	ListPrice	ListPriceRank
314	Road-150 Red, 56	\$3,578.27	1
313	Road-150 Red, 52	\$3,578.27	1
312	Road-150 Red, 48	\$3,578.27	1
311	Road-150 Red, 44	\$3,578.27	1
310	Road-150 Red, 62	\$3,578.27	1
347	Mountain-100 Si...	\$3,399.99	2
346	Mountain-100 Si...	\$3,399.99	2
345	Mountain-100 Si...	\$3,399.99	2
344	Mountain-100 Si...	\$3,399.99	2
351	Mountain-100 B...	\$3,374.99	3
350	Mountain-100 B...	\$3,374.99	3
349	Mountain-100 B...	\$3,374.99	3
348	Mountain-100 B...	\$3,374.99	3
380	Road-250 Black, ...	\$2,443.35	4
378	Road-250 Black, ...	\$2,443.35	4



Nesting row contexts

When you nest row contexts on the same table, you can use a variable to save the value of the outer row context

```
Products[RankOnPrice] =  
VAR CurrentListPrice = Products[ListPrice]  
RETURN  
COUNTRows (  
    FILTER (  
        Products,  
        Products[ListPrice] > CurrentListPrice  
    )  
) + 1
```

Row context of the calculated column

Get ListPrice from the outer row context

Row context of the FILTER function



Nesting row contexts

As an alternative, in versions of DAX that do not support variables, you can use the EARLIER function

```
Products[RankOnPrice] =  
  
COUNTRows (  
    FILTER (  
        Products,  
        Products[ListPrice] > EARLIER ( Products[ListPrice] )  
    )  
) + 1
```

Row context of the
calculated column

Row context of the
FILTER function

Get ListPrice from the
outer row context



Computing the correct rank

The correct solution requires to rank over the different prices, not the different products. ALL with a column becomes very handy here.


```
Products[RankOnPrice] =  
VAR CurrentListPrice = Products[ListPrice]  
VAR AllPrices = ALL ( Products[ListPrice] )  
RETURN  
COUNTROWS (  
FILTER (  
AllPrices,  
Products[ListPrice] > CurrentListPrice  
)  
) + 1
```



Evaluation contexts and relationships



Filters and relationships



- Do contexts interact with relationships?
 - Row Context
 - Filter Context
 - One side
 - Many side



Row context – multiple tables

Row Context does not propagate over relationships

										f _x	=Orders[Amount] * (1 - Channels[Discount])		
channel	Color	Size	Quantity	Price	Amount					DiscountedAmount			Ad
core	Red	Large	1	15	15					#ERROR			
core	Red	Small	2	10	20					#ERROR			
core													
core													
internet													
internet													
internet													
internet	Green	Small	120	1	120								

A single value for column 'Discount' in table 'Channels' cannot be determined. This can happen when a measure formula refers to a column that contains many values without specifying an aggregation such as min, max, count, or sum to get a single result.



RELATED

RELATED (table[column])

- Opens a new row context on the target table
- Following relationships

	City	Cha...	Color	Size	Quantity	Price	Amount	DiscountedAmount
	Paris	Store	Red	Large	1	15	15	14.25
	Paris	Store	Red	Small	2	13	26	24.7
	Torino	Store	Green	Large	4	11	44	41.8
	New York	Store	Green	Small	8	9	72	68.4
		Internet	Red	Large	16	7	112	100.8
		Internet	Red	Small	32	5	160	144
		Internet	Green	Large	64	3	192	172.8
		Internet	Green	Small	128	1	128	115.2



RELATEDTABLE

1. RELATEDTABLE (table)

- Filters the parameter table
- Returns only rows related with the current one

1. It is the companion of RELATED

[OrdersCount]	f _x	=COUNTROWS(RELATEDTABLE(Orders))	
Channel	Discount	OrdersCount	Add Column
Internet	0.1	4	
Store	0.05	4	



Filter context – many tables

The screenshot illustrates the concept of filter context in PowerPivot across multiple tables.

PowerPivot Field List:

- Orders:** Color (checked), Size (checked), Quantity (unchecked), Price (unchecked), Amount (unchecked), DiscountedAmount (checked), CalcAmount (unchecked).
- Channels:** Channel (checked), Discount (unchecked).
- Cities:** City (unchecked), Country (unchecked), Continent (checked).

Tables and Filter Context:


- Continent Table:** Row Labels: Europe (selected), North America.
- Channel Table:** Row Labels: Store (selected), Internet.
- Country Table:** Row Labels: Paris (France, Europe), New York (USA, North America), Torino (Italy, Europe), Madrid (Spain, Europe).
- Product Table:** Row Labels: Paris (Store, Red, Large), Paris (Store, Red, Small), Torino (Store, Green, Large), New York (Store, Green, Small), Internet (Red, Large), Internet (Red, Small), Internet (Green, Large), Internet (Green, Small). The table shows calculated values for Price, Amount, and DiscountedAmount.
- PivotTable:** Sum of Discounted/Column Labels. Row Labels: Large, Small, Grand Total. Data:

	Large	Small	Grand Total
Green	41.8		41.8
Red	14.25	24.7	38.95
Grand Total	56.05	24.7	80.75

Red arrows indicate the flow of filter context from the Continent and Channel tables into the Product table. A blue arrow points from the PivotTable's Grand Total back to the Product table's Grand Total row.



Filters and relationships





Bidirectional cross-filter

1. Choose the propagation of the filter context
 - Single: one to many propagation
 - Both: filter context propagates both ways
1. Not available in Excel, only Analysis Services and Power BI
1. Beware of several details
 - Performance degradation
 - Filtering is active when the “many” side is cross-filtered, numbers might be hard to read for certain measures
 - Ambiguity might appear in the model



The most important DAX function

CALCULATE



CALCULATE syntax

Filters are evaluated in the outer filter context, then combined together in AND, and finally used to build a new filter context into which DAX evaluates the expression.

```
CALCULATE (
    Expression,
    Filter1,
    ...
    Filtern )
```

Repeated many times, as needed

The filter parameters are used to modify the existing filter context. They can add, remove or change existing filter.



CALCULATE examples

Compute the sum of sales where the price is greater than \$100.00.

```
NumOfBigSales :=  
  
CALCULATE (  
    SUMX ( Sales, Sales[Quantity] * Sales[Net Price] ),  
    Sales[Net Price] > 100  
)
```

Filter and SUM are
on the same table.
You can obtain the same
result using FILTER.



Filters are tables

Each filter is a table. Boolean expressions are nothing but shortcuts for table expressions.

```
CALCULATE (
    [Sales Amount],
    Sales[Net Price] > 100
)
```

Is equivalent to

```
CALCULATE (
    [Sales Amount],
    FILTER (
        ALL ( Sales[Net Price] ),
        Sales[Net Price] > 100
    )
)
```



Let's learn CALCULATE by using some demo

CALCULATE Examples



CALCULATE examples

Compute the sales amount for all of the product colors, regardless of the user selection.

```
SalesAllColors :=
```

```
    CALCULATE (
        [Sales Amount],
        ALL ( Product[Color] )
    )
```

The condition is the list of
acceptable values



CALCULATE examples

Compute the sales amount for red products, regardless of user selection for color. The filter context is applied on the entire model, products filter the Sales table, too.

```
SalesRedProducts :=
```

```
    CALCULATE (
        [Sales Amount],
        Product[Color] = "Red"
    )
```

Filter and SUM are on different tables.
Filter happens because of filter context propagation.



CALCULATE examples

Compute the sales amount for red and blue products, within the user selection for color.

```
SalesTrendyColors :=  
  
CALCULATE (  
    [Sales Amount],  
    KEEPFILTERS ( Product[Color] IN { "Red", "Blue" } )  
)
```

KEEPFILTERS keeps the previous filter, so that the new filter does not override the previous one



What is a filter context?

Compute the sales amount for red and blue products, within the user selection for color.

```
CALCULATE (
    ...,
    Product[Color] IN { "Red", "Black" },
    FILTER (
        ALL ( Date[Year], Date[Month] ),
        OR (
            AND ( Date[Year] = 2006, Date[Month] = "December" ),
            AND ( Date[Year] = 2007, Date[Month] = "January" )
        )
    )
)
```




Year	Month	Color
2006	December	Red
2007	January	Black

Filters are tables.
You can use any table
function to create a filter in
CALCULATE



Filter context definition



Year	Month	Color
2006	December	Red
2007	January	Black

Tuple: value for a set of columns

Filter: table of tuples

Filter context: set of filters



Multiple conditions in CALCULATE


Multiple filter parameters in CALCULATE are intersected, generating a new filter context that uses both filters at the same time.

```
CALCULATE (
    ...,
    Product[Color] IN { "Red", "Black" },
    FILTER (
        ALL ( Date[Year], Date[Month] ),
        OR (
            AND ( Date[Year] = 2006, Date[Month] = "December" ),
            AND ( Date[Year] = 2007, Date[Month] = "January" )
        )
    )
)
```



Intersection of filter context

Used by CALCULATE to put filters in AND






Overwriting filter contexts

Nested CALCULATE do not intersect filters, they use another operator, called OVERWRITE. In fact, the Yellow/Black filter wins against the Black/Blue one, being the innermost. Yellow/Black overwrites Black/Blue.

```
CALCULATE (
    CALCULATE (
        ...,
        Product[Color] IN { "Yellow", "Black" }
    ),
    Product[Color] IN { "Black", "Blue" }
)
```






KEEPFILTERS

KEEPFILTERS retains the previous filters, instead of replacing them.

```
CALCULATE (
    CALCULATE (
        ...
        KEEPFILTERS ( Product[Color] IN { "Yellow", "Black" } )
    ),
    Product[Color] IN { "Black", "Blue" }
)
```





CALCULATE operators

- o Overwrite a filter context at the individual columns level
- o Remove previously existing filters (ALL)
- o Add filters (KEEPFILTERS)
- o In DAX you work by manipulating filters with the following internal operators:
 - INTERSECT (multiple filters in CALCULATE)
 - OVERWRITE (nested CALCULATE)
 - REMOVEFILTERS (using ALL)
 - ADDFILTER (using KEEPFILTERS)



KEEPFILTERS might be required, don't filter the table!

KEEPFILTERS is required to keep the same semantics of the table filter. You obtain a similar result by filtering the entire table. Filtering a table is a very bad practice.

```
CALCULATE (
    [Sales Amount],
    KEEPFILTERS (
        FILTER (
            ALL ( Sales[Quantity], Sales[Price] ),
            Sales[Quantity] * Sales[Net Price] > 1000
        )
    )
)
```

```
CALCULATE (
    [Sales Amount],
    FILTER (
        Sales,
        Sales[Quantity] * Sales[Net Price] > 1000
    )
)
```

NEVER filter a
table!



Aggregators in compact syntax

Aggregators in compact syntax are evaluated in the outer filter context

```
CALCULATE (
    [Sales Amount],
    Sales[Quantity] < SUM ( Sales[Quantity] ) / 100
)
```

Using variables, however, makes the code more readable

```
VAR TotalQuantity = SUM ( Sales[Quantity] )
RETURN
CALCULATE (
    [Sales Amount],
    Sales[Quantity] < TotalQuantity / 100
)
```



Variables and evaluation contexts

Variables are computed in the evaluation where they are defined, not in the one where they are used. CALCULATE cannot modify the value of a variable because it is already computed.

```
WrongRatio :=  
  
VAR  
    Amt = [Sales Amount]  
RETURN  
    DIVIDE (  
        Amt,  
        CALCULATE ( Amt, ALL ( Sales ) )  
    )
```

Result is always 1



One more feature of CALCULATE

Context transition



Context transition

- o Calculate performs another task
- o If executed inside a row context
 - It takes the row context
 - Transforms it into an equivalent filter context
 - Applies it to the data model
 - Before computing its expression
- o Very important and useful feature
 - Better to learn it writing some code...



Context transition

Sales		
Product	Quantity	Net Price
A	1	11.00
B	2	25.00
A	2	10.99

Row Context

```
Test :=  
SUMX (   
    Sales,  
    CALCULATE ( SUM ( Sales[Quantity] ) )  
)
```

Filter Context

Product	Quantity	Net Price
A	1	11.00

SUMX Iteration

Row Iterated	Sales[Quantity] Value	Row Result
1	1	1
2	2	2
3	2	2

The result of
SUMX is 5



Unexpected results if there are duplicated rows

Sales

Product	Quantity	Net Price
A	1	11.00
B	2	25.00
B	2	25.00

Row Context

```
Test :=  
SUMX (   
    Sales,  
    CALCULATE ( SUM ( Sales[Quantity] ) )  
)
```

Filter Context

SUMX Iteration

Row Iterated	Sales[Quantity] Value	Row Result
1	1	1
2	2	4
3	2	4

The result of
SUMX is 9

Product	Quantity	Net Price
B	2	25.00



Some notes on context transition

- It is invoked by CALCULATE
- It is expensive: don't use it iterating large tables
- It does not filter one row, it filters all the identical rows
- It creates a filter context out of a row context
- It happens whenever there is a row context
- It transforms all the row contexts, not only the last one
- Row contexts are no longer available in CALCULATE



Automatic CALCULATE

Whenever a measure is invoked, an automatic CALCULATE is added around the measure. This is the reason why using [Measure] and Table[Column] as a standard is a best practice.

```
SUMX (  
    Orders,  
    [Sales Amount]  
)  
SUMX (  
    Orders,  
    CALCULATE ( [Sales Amount] )  
)
```



Time to start thinking in DAX

Working with iterators



Computing max daily sales

- What is the maximum amount of sales in one day?
- MAX is not enough
- Consolidate the daily amount and then find the maximum value

Year	Sales Amount	Max Daily Sales
□ CY 2007	11,309,946.12	126,742.18
□ January 2007	794,248.24	92,244.07
01/02/2007	48,646.02	48,646.02
01/03/2007	92,244.07	92,244.07
01/04/2007	13,950.29	13,950.29
01/05/2007	62,050.83	62,050.83
01/07/2007	23,305.63	23,305.63
01/09/2007	20,543.35	20,543.35
01/10/2007	6,565.56	6,565.56
01/11/2007	22,693.05	22,693.05
01/12/2007	16,251.63	16,251.63
01/13/2007	11,315.05	11,315.05
01/15/2007	58,224.87	58,224.87
01/16/2007	45,595.65	45,595.65



MIN-MAX sales per customer

Iterators can be used to compute values at a different granularity than the one set in the report.

```
MinSalesPerCustomer :=  
    MINX ( Customer, [Sales Amount] )
```

```
MaxSalesPerCustomer :=  
    MAXX ( Customer, [Sales Amount] )
```



Useful iterators

There are many useful iterators, they all behave the same way: iterate on a table, compute an expression and aggregate its value.

```
MAXX  
MINX  
AVERAGEX  
SUMX  
PRODUCTX  
CONCATENATEXVARX.P | .S  
STDEVX.P | .S  
MEDIANX  
PERCENTILEX.EXC | .INC  
GEOMEANX
```



Useful iterators

There are many useful iterators, they all behave the same way: iterate on a table, compute an expression and aggregate its value.

```
MAXX  
MINX  
AVERAGEX  
SUMX  
PRODUCTX  
CONCATENATEXVARX.P | .S  
STDEVX.P | .S  
MEDIANX  
PERCENTILEX.EXC | .INC  
GEOMEANX
```



Probably the most important table in your model

Building a date table



Date table

1. Time intelligence needs a date table

- Built in DAX, Power Query, SQL
- DAX example: <https://www.sqlbi.com/tools/dax-date-template/>

2. Date table properties

- All dates should be present –no gaps
- From 1° of January, to 31° of December
- Or for the fiscal calendar including full months
- Otherwise time intelligence DAX functions do not work



Auto date/time and column variations

In Power BI the auto date/time setting automatically creates one date table for each date column in the model.

It is not a best practice, unless you are using a very simple model. Columns in these auto-created tables can be accessed through “column variations”.

```
Sales Amount YTD :=  
TOTALYTD (  
    'Sales'[Sales Amount],  
    'Sales'[Order Date].[Date]  
)
```

Column variation

Year	Quarter	Month	Day	Sales Amount	Sales Amount YTD
2009	Qtr 1	January	1	2,198.95	2,198.95
			2	1,325.89	3,524.84
			3	1,775.52	5,300.35
			4	2,167.90	7,468.25
			5	511.70	7,979.95
			6	907.89	8,887.84
			7	332.37	9,220.21
			8	4,605.52	13,825.73
			9	4,442.14	18,267.87
			10	82.70	18,350.58
			11	60.44	18,411.01
			12	1,771.18	20,182.19



CALENDARAUTO

Automatically creates a calendar table based on the database content.

Optionally you can specify the last month (for fiscal years).

```
--  
-- The parameter is the last month  
-- of the fiscal year  
  
= CALENDARAUTO (  
    6  
)
```

Beware: CALENDARAUTO uses all the dates in your model, excluding only calculated columns and tables



CALENDAR

Returns a table with a single column named “Date” containing a contiguous set of dates in the given range, inclusive.

```
CALENDAR (DATE ( 2005, 1, 1 ),  
DATE ( 2015, 12, 31 )  
)
```

```
CALENDAR (  
MIN ( Sales[Order Date] ),  
MAX ( Sales[Order Date] )  
)
```



Mark as date table

1. Need to mark the calendar as date table
- 2 . Set the column containing the date
- 3 . Needed to make time intelligence work if the relationship does not use a Date column
4. Multiple tables can be marked as date table
- 5 . Used by client tools as metadata information



Set sorting options

- Month names do not sort alphabetically
 - April is not the first month of the year
- Use Sort By Column
- Set all sorting options in the proper way
- Beware of sorting granularity
 - 1:1 between names and sort keys



Using multiple dates

- Date is often a role dimension
 - Many roles for a date
 - Many date tables
- How many date tables?
 - Try to use only one table
 - Use many, only if needed by the model
 - Many date tables lead to confusion
 - And issues when slicing
- Use proper naming convention



Time intelligence functions

Time intelligence in DAX



What is time intelligence?

1. Many different topics in one name

- Year To Date
- Quarter To Date
- Running Total
- Same period previous year
- Working days computation
- Fiscal Year



Aggregations over time

- Many useful aggregations
 - YTD: Year To Date
 - QTD: Quarter To Date
 - MTD: Month To Date
- They all need a date table
- And some understanding of CALCULATE



Sales 2015 up to 05-15 (v1)

Using CALCULATE you can filter the dates of the period to summarize.

```
SalesAmount20150515 :=  
CALCULATE (  
SUM ( Sales[SalesAmount] ),  
FILTER (  
ALL ( 'Date'[Date] ),  
AND ( 'Date'[Date] >= DATE ( 2015, 1, 1 ),  
'Date'[Date] <= DATE ( 2015, 5, 15 )  
)  
)  
)
```



Sales 2015 up to 05-15 (v2)

You can replace FILTER with DATESBETWEEN.

The result is always a table with a column.

```
SalesAmount20150515 :=  
CALCULATE (  
SUM ( Sales[SalesAmount] ),  
DATESBETWEEN (  
'Date'[Date],  
DATE ( 2015, 1, 1 ),DATE ( 2015, 5, 15 )  
)  
)
```



Sales Year-To-Date (v1)

Replace the static dates using DAX expressions that retrieve the last day in the current filter.

```
SalesAmountYTD:=  
CALCULATE (  
SUM ( Sales[SalesAmount] ),  
DATESBETWEEN (  
'Date'[Date],  
DATE ( YEAR ( MAX ( 'Date'[Date] ) ) ), 1, 1 ),MAX (   
'Date'[Date] )  
)  
)
```



Year to date (Time Intelligence)

Replace the static dates using DAX expressions that retrieve the last day in the current filter.

```
SalesAmountYTD:=  
CALCULATE (  
SUM ( Sales[SalesAmount] ),  
DATESBETWEEN (  
'Date'[Date],  
DATE ( YEAR ( MAX ( 'Date'[Date] ) ) ), 1, 1 ),MAX (   
'Date'[Date] )  
)  
)
```



Year to date: the easy way

TOTALYTD: the “DAX for dummies” version.

It hides the presence of CALCULATE, so we suggest not to use it.

```
SalesAmountYTD :=
```

```
TOTALYTD (
    SUM ( Sales[SalesAmount] ),
    'Date'[Date]
)
```



Handling fiscal year

The last, optional, parameter is the end of the fiscal year.


Default: 12-31 (or 31/12 –you can use any format regardless of locale settings).

```
SalesAmountYTD:=
```

```
TOTALYTD (SUM ( Sales[SalesAmount] ), 'Date'[Date], "06-  
30")
```

```
SalesAmountYTD:=
```

```
CALCULATE (SUM ( Sales[SalesAmount] ), DATESYTD (   
'Date'[Date], "06-30" ))
```



Thank you

Follow me and Growtechie

@Ramakrushnamohapatra
@growtechieind